

Aplicar filtros de expresión booleana

Un filtro de expresión booleana es una expresión que se evalúa como TRUE o FALSE.

Abrimos el archivo LABORATORIO 3.pbix A continuación, agregue la siguiente medida a la tabla **Sales** que filtra la medida **Revenue** mediante un filtro de expresión booleana para los productos rojos.

Revenue Red = CALCULATE([Revenue], 'Product'[Color] = "Red")

Agregue la medida **Revenue Red** al objeto visual de tabla que se encuentra en la **Página 1** del informe.

Region	Revenue	Revenue Red
Australia	\$10,655,335.96	\$2,681,324.79
Canada	\$16,355,770.46	\$3,573,412.99
Central	\$7,909,009.01	\$1,585,997.34
France	\$7,251,555.65	\$1,051,014.15
Germany	\$4,878,300.38	\$670,607.30
Northeast	\$6,939,374.48	\$1,876,016.33
Northwest	\$16,084,942.55	\$2,292,905.61
Southeast	\$7,879,655.07	\$1,457,221.07
Southwest	\$24,184,609.60	\$5,345,637.47
United Kingdom	\$7,670,721.04	\$1,063,753.75
Total	\$109,809,274.20	\$21,597,890.81

En el siguiente ejemplo, la medida siguiente filtra la medida **Revenue** por varios colores. Observe el uso del operador IN seguido de una lista de valores de color.

Revenue Red or Blue = CALCULATE([Revenue], 'Product'[Color] IN {"Red", "Blue"})

EJERCICIO – Vamos a filtrar la medida **Revenue por productos caros. Los productos costosos son aquellos con un precio de venta superior a 1000 USD. Crea una medida llamada Revenue Expensive Products que nos devuelva el Revenue de dichos productos**

Aplicar filtros de expresión de tabla

Un filtro de expresión de tabla aplica un objeto de tabla como un filtro. Podría ser una referencia a una tabla de modelo; sin embargo, es probable que sea una función de DAX que devuelve un objeto de tabla.

Normalmente, usará la función FILTER de DAX para aplicar condiciones de filtro complejas, incluidas las que no se pueden definir mediante una expresión de filtro booleana. La función FILTER se clasifica como una función de iterador, por lo que se pasaría en una tabla o una expresión de tabla, y una expresión para evaluar para cada fila de esa tabla.

La función FILTER devuelve un objeto de tabla con exactamente la misma estructura que la tabla pasada. Sus filas son un subconjunto de esas filas que se pasaron, que son las filas donde la expresión se evaluó como TRUE.

En el ejemplo siguiente, se muestra una expresión de filtro de tabla que utiliza la función FILTER:

```
Revenue High Margin Products =  
CALCULATE(  
    [Revenue],  
    FILTER(  
        'Product',  
        'Product'[List Price] > 'Product'[Standard Cost] * 2  
    )  
)
```

En este ejemplo, la función FILTER filtra todas las filas de la tabla Product que se encuentran en el contexto de filtro. Cada fila de un producto cuyo precio de lista supera el doble de su costo estándar se muestra como una fila de la tabla filtrada. Por lo tanto, la medida **Revenue** se evalúa para todos los productos devueltos por la función FILTER.

Todas las expresiones de filtro que se pasan a la función CALCULATE son expresiones de filtro de tabla. Una expresión de filtro booleana es una notación abreviada para mejorar la experiencia de lectura y escritura. Internamente, Microsoft Power BI traduce las expresiones de filtro booleanas en expresiones de filtro de tabla, que es cómo traduce la definición de medida **Revenue Red**.

```
Revenue Red =  
CALCULATE(  
    [Revenue],  
    FILTER(  
        'Product',  
        'Product'[Color] = "Red"  
    )  
)
```

Quitar filtros

Use la función REMOVEFILTERS de DAX como una expresión de filtro CALCULATE para quitar filtros del contexto de filtro. Puede quitar filtros de una o varias columnas o de todas las columnas de una única tabla.

La función REMOVEFILTERS es relativamente nueva. En las versiones anteriores de DAX, los filtros se quitaban con las variantes o la función **ALL** de DAX, incluidas las funciones **ALLEXCEPT** y **ALLNOBLANKROW** de DAX. Estas funciones se comportan como modificadores de filtro y como funciones que devuelven objetos de tabla de valores distintos. Estas funciones se mencionan ahora porque es probable que encuentre ejemplos de documentación y fórmulas que las usen para quitar filtros.

En el ejemplo siguiente, agregará una medida a la tabla **Sales** que evalúa la medida **Revenue**, pero lo hace quitando los filtros de la tabla **Sales Territory**. Dé formato a la medida como moneda con dos números decimales.

Revenue Total Region = CALCULATE([Revenue], REMOVEFILTERS('Sales Territory'))

Ahora, agregue la medida **Revenue Total Region** al objeto visual de matriz que se encuentra en la **página 2** del informe. El objeto visual de matriz se agrupará en tres columnas de la tabla Sales Territory en las filas: **Group**, **Country** y **Region**.

Reseller Revenue				
Group	Country	Region	Revenue	Revenue Total Region
Corporate HQ	Corporate HQ	Corporate HQ		\$80,450,596.98
		Total		\$80,450,596.98
	Total			\$80,450,596.98
Europe	France	France	\$4,607,537.94	\$80,450,596.98
		Total	\$4,607,537.94	\$80,450,596.98
	Germany	Germany	\$1,983,988.04	\$80,450,596.98
		Total	\$1,983,988.04	\$80,450,596.98
	United Kingdom	United Kingdom	\$4,279,008.83	\$80,450,596.98
		Total	\$4,279,008.83	\$80,450,596.98
	Total		\$10,870,534.80	\$80,450,596.98
North America	Canada	Canada	\$14,377,925.60	\$80,450,596.98
		Total	\$14,377,925.60	\$80,450,596.98
	United States	Central	\$7,906,008.18	\$80,450,596.98
		Northeast	\$6,932,842.01	\$80,450,596.98
		Northwest	\$12,435,076.00	\$80,450,596.98
		Southeast	\$7,867,416.23	\$80,450,596.98
		Southwest	\$18,466,458.79	\$80,450,596.98
		Total	\$53,607,801.21	\$80,450,596.98
	Total		\$67,985,726.81	\$80,450,596.98
Pacific	Australia	Australia	\$1,594,335.38	\$80,450,596.98
		Total	\$1,594,335.38	\$80,450,596.98
	Total		\$1,594,335.38	\$80,450,596.98
Total			\$80,450,596.98	\$80,450,596.98

Tenga en cuenta todos los valores de **Revenue Total Region** son los mismos. Se trata del valor de los ingresos totales.

Aunque este resultado no resulta útil, cuando se usa como un denominador en una proporción, calcula un porcentaje del total general. Por tanto, ahora sobrescribirá la definición de medida **Revenue Total Region** con la siguiente definición. (Esta nueva definición cambia el nombre de la medida y declara dos variables. Asegúrese de dar formato a la medida como un porcentaje con dos posiciones decimales).

```
Revenue % Total Region =
VAR CurrentRegionRevenue = [Revenue]
VAR TotalRegionRevenue =
    CALCULATE(
        [Revenue],
        REMOVEFILTERS('Sales Territory')
    )
RETURN
    DIVIDE(
        CurrentRegionRevenue,
        TotalRegionRevenue
    )
```

Compruebe que el objeto visual de matriz ahora muestra los valores de **Revenue % Total Region**.

Reseller Revenue

Group	Country	Region	Revenue	Revenue % Total Region
Europe	France	France	\$4,607,537.94	5.73%
		Total	\$4,607,537.94	5.73%
	Germany	Germany	\$1,983,988.04	2.47%
		Total	\$1,983,988.04	2.47%
	United Kingdom	United Kingdom	\$4,279,008.83	5.32%
		Total	\$4,279,008.83	5.32%
Total		\$10,870,534.80	13.51%	
North America	Canada	Canada	\$14,377,925.60	17.87%
		Total	\$14,377,925.60	17.87%
	United States	Central	\$7,906,008.18	9.83%
		Northeast	\$6,932,842.01	8.62%
		Northwest	\$12,435,076.00	15.46%
		Southeast	\$7,867,416.23	9.78%
		Southwest	\$18,466,458.79	22.95%
		Total	\$53,607,801.21	66.63%
	Total		\$67,985,726.81	84.51%
Pacific	Australia	Australia	\$1,594,335.38	1.98%
		Total	\$1,594,335.38	1.98%
	Total		\$1,594,335.38	1.98%
Total		\$80,450,596.98	100.00%	

Ahora, creará otra medida, pero esta vez, calculará la proporción de ingresos para una región dividida por los ingresos de su país.

En la definición de medida siguiente, observe cómo puede borrar o quitar un filtro de una columna. En la lógica de DAX, se trata de un cambio pequeño y sutil que se realiza en la fórmula de medida **Revenue % Total Region**: La función REMOVEFILTERS ahora quita los filtros de la columna **Region**, en lugar de todas las columnas de la tabla **Sales Territory**.

```
Revenue % Total Country =
VAR CurrentRegionRevenue = [Revenue]
VAR TotalCountryRevenue =
    CALCULATE(
        [Revenue],
        REMOVEFILTERS('Sales Territory'[Region])
    )
RETURN
    DIVIDE(
        CurrentRegionRevenue,
        TotalCountryRevenue
    )
```

Agregue la medida **Revenue % Total Country** a la tabla **Sales** y, después, aplíquelo formato como un porcentaje con dos posiciones decimales. Agregue la nueva medida al objeto visual de matriz.

Reseller Revenue

Group	Country	Region	Revenue	Revenue % Total Region	Revenue % Total Country
Europe	France	France	\$4,607,537.94	5.73%	100.00%
		Total	\$4,607,537.94	5.73%	100.00%
	Germany	Germany	\$1,983,988.04	2.47%	100.00%
		Total	\$1,983,988.04	2.47%	100.00%
	United Kingdom	United Kingdom	\$4,279,008.83	5.32%	100.00%
		Total	\$4,279,008.83	5.32%	100.00%
	Total		\$10,870,534.80	13.51%	100.00%
North America	Canada	Canada	\$14,377,925.60	17.87%	100.00%
		Total	\$14,377,925.60	17.87%	100.00%
	United States	Central	\$7,906,008.18	9.83%	14.75%
		Northeast	\$6,932,842.01	8.62%	12.93%
		Northwest	\$12,435,076.00	15.46%	23.20%
		Southeast	\$7,867,416.23	9.78%	14.68%
		Southwest	\$18,466,458.79	22.95%	34.45%
		Total	\$53,607,801.21	66.63%	100.00%
	Total		\$67,985,726.81	84.51%	100.00%
Pacific	Australia	Australia	\$1,594,335.38	1.98%	100.00%
		Total	\$1,594,335.38	1.98%	100.00%
	Total		\$1,594,335.38	1.98%	100.00%
Total			\$80,450,596.98	100.00%	100.00%

Observe que todos los valores, excepto los valores de las regiones de Estados Unidos, son del 100 %. La razón es que, en la compañía Adventure Works, Estados Unidos tiene regiones, mientras que todos los demás países no tienen.

Los modelos tabulares no admiten jerarquías desequilibradas, que son jerarquías con profundidades variables. Por lo tanto, se trata de un enfoque de diseño común para repetir valores primarios u otros elementos antecesores en los niveles inferiores de la jerarquía. Por ejemplo, Australia no tiene ninguna

región, por lo que el valor del país y región se repite como el nombre de la región. Siempre es mejor almacenar un valor significativo en lugar de BLANK.

EJERCICIO – Define la medida **Revenue % Total Group que nos permita conocer el % de revenue que tiene cada país dentro de su grupo. Agregala a la tabla, debes de obtener una visualización similar a la siguiente:**

Reseller Revenue

Group	Country	Region	Revenue	Revenue % Total Region	Revenue % Total Country	Revenue % Total Group
Europe	France	France	\$4,607,537.94	5.73%	100.00%	42.39%
		Total	\$4,607,537.94	5.73%	100.00%	42.39%
	Germany	Germany	\$1,983,988.04	2.47%	100.00%	18.25%
		Total	\$1,983,988.04	2.47%	100.00%	18.25%
	United Kingdom	United Kingdom	\$4,279,008.83	5.32%	100.00%	39.36%
		Total	\$4,279,008.83	5.32%	100.00%	39.36%
	Total		\$10,870,534.80	13.51%	100.00%	100.00%
North America	Canada	Canada	\$14,377,925.60	17.87%	100.00%	21.15%
		Total	\$14,377,925.60	17.87%	100.00%	21.15%
	United States	Central	\$7,906,008.18	9.83%	14.75%	11.63%
		Northeast	\$6,932,842.01	8.62%	12.93%	10.20%
		Northwest	\$12,435,076.00	15.46%	23.20%	18.29%
		Southeast	\$7,867,416.23	9.78%	14.68%	11.57%
		Southwest	\$18,466,458.79	22.95%	34.45%	27.16%
		Total	\$53,607,801.21	66.63%	100.00%	78.85%
	Total		\$67,985,726.81	84.51%	100.00%	100.00%
Pacific	Australia	Australia	\$1,594,335.38	1.98%	100.00%	100.00%
		Total	\$1,594,335.38	1.98%	100.00%	100.00%
	Total		\$1,594,335.38	1.98%	100.00%	100.00%
Total			\$80,450,596.98	100.00%	100.00%	100.00%

Conservar filtros

Puede usar la función **KEEPFILTERS** de DAX como una expresión de filtro en la función **CALCULATE** para conservar los filtros.

Para observar cómo llevar a cabo esta tarea, cambie a la **página 1** del informe. A continuación, modifique la definición de medida **Revenue Red** para usar la función **KEEPFILTERS**.

```
Revenue Red =  
CALCULATE(  
    [Revenue],  
    KEEPFILTERS('Product'[Color] = "Red")  
)
```

Color	Revenue	Revenue Red
Black	\$38,236,124.06	
Blue	\$9,602,850.97	
Multi	\$649,030.25	
NA	\$1,099,303.91	
Red	\$21,597,890.81	\$21,597,890.81
Silver	\$19,777,339.95	
Silver/Black	\$147,483.91	
White	\$29,745.13	
Yellow	\$18,669,505.22	
Total	\$109,809,274.20	\$21,597,890.81

En el objeto visual de tabla, observe que solo existe un valor **Revenue Red**. La razón es que la expresión de filtro booleana conserva los filtros existentes en la columna **Color** de la tabla **Product**. El motivo por el que los colores distintos del rojo son BLANK es porque los contextos de filtro y las expresiones de filtro se combinan para estos dos filtros. Los colores negro y rojo se intersecan y, dado que ambos no pueden ser TRUE al mismo tiempo, la expresión no se filtra por ninguna fila de producto. Solo es posible que ambos filtros rojos puedan ser TRUE al mismo tiempo, lo que explica por qué se muestra un valor **Revenue Red**.

Usar relaciones inactivas

Una relación de modelo inactiva solo puede propagar los filtros cuando la función USERELATIONSHIP de DAX se pasa como una expresión de filtro a la función CALCULATE. Cuando se usa esta función para interactuar con una relación inactiva, la relación activa se convierte automáticamente en inactiva.

Revise un ejemplo de una definición de medida que usa una relación inactiva para calcular la medida **Revenue** por fechas de envío:

```
Revenue Shipped =
CALCULATE (
    [Revenue],
    USERELATIONSHIP('Date'[DateKey], Sales[ShipDateKey])
)
```

Examinar el contexto de filtro

La función VALUES de DAX permite que las fórmulas determinen qué valores se encuentran en el contexto de filtro.

La sintaxis de la función VALUES es la siguiente:

VALUES(<TableNameOrColumnName>)

La función requiere pasar una referencia de tabla o una referencia de columna. Cuando se pasa una referencia de tabla, devuelve un objeto de tabla con las mismas columnas que contienen filas para lo que está en el contexto de filtro. Cuando se pasa una referencia de columna, devuelve una tabla de una sola columna con valores únicos que se encuentran en el contexto del filtro.

La función siempre devuelve un objeto de tabla y es posible que una tabla contenga varias filas. Por lo tanto, para comprobar si un valor específico está en el contexto de filtro, la fórmula debe comprobar primero que la función VALUES devuelve una sola fila. Dos funciones pueden ayudarle a realizar esta tarea: las funciones HASONEVALUE y SELECTEDVALUE de DAX.

En el ejemplo siguiente, usará la función HASONEVALUE. Agregue la siguiente medida, que calcula la comisión de ventas, a la tabla **Sales**. Tenga en cuenta que, en Adventure Works, la tarifa de la comisión es el 10 % de los ingresos de todos los países y regiones, excepto Estados Unidos. En Estados Unidos, el personal de ventas gana una comisión del 15 %. Dé formato a la medida como moneda con dos posiciones decimales y, luego, agréguela a la tabla que se encuentra en la **Página 3** del informe.

```
Sales Commission =  
[Revenue]  
    * IF(  
        HASONEVALUE('Sales Territory'[Country]),  
        IF(  
            VALUES('Sales Territory'[Country]) = "United States",  
            0.15,  
            0.1  
        )  
    )  
)
```


Region	Revenue	Sales Commission
Australia	\$10,655,335.96	\$1,065,533.60
Canada	\$16,355,770.46	\$1,635,577.05
Central	\$7,909,009.01	\$1,186,351.35
France	\$7,251,555.65	\$725,155.56
Germany	\$4,878,300.38	\$487,830.04
Northeast	\$6,939,374.48	\$1,040,906.17
Northwest	\$16,084,942.55	\$2,412,741.38
Southeast	\$7,879,655.07	\$1,181,948.26
Southwest	\$24,184,609.60	\$3,627,691.44
United Kingdom	\$7,670,721.04	\$767,072.10
Total	\$109,809,274.20	

Tenga en cuenta que el resultado total de **Sales Commission** es BLANK. La razón es que hay varios valores en el contexto de filtro para la columna **Country** de la tabla **Sales Territory**. En este caso, la función HASONESVALUE devuelve FALSE, lo que da lugar a que se multiplique la medida **Revenue** por BLANK (un valor multiplicado por BLANK es BLANK). Para generar un total, debe usar una función de iterador, que se explica más adelante en este módulo.

Vuelva a la **Página 2** del informe y, a continuación, modifique la definición de la medida **Revenue % Total Country** para comprobar que la columna **Region** de la tabla **Sales Territory** está en el ámbito. Si no está en el ámbito, el resultado de la medida debe ser BLANK.

```

Revenue % Total Country =
VAR CurrentRegionRevenue = [Revenue]
VAR TotalCountryRevenue =
    CALCULATE(
        [Revenue],
        REMOVEFILTERS('Sales Territory'[Region])
    )
RETURN
    IF(
        ISINSCOPE('Sales Territory'[Region]),
        DIVIDE(
            CurrentRegionRevenue,
            TotalCountryRevenue
        )
    )

```

Reseller Revenue						
Group	Country	Region	Revenue	Revenue % Total Region	Revenue % Total Country	Revenue % Total Group
Europe	France	France	\$4,607,537.94	5.73%	100.00%	42.39%
		Total	\$4,607,537.94	5.73%		42.39%
	Germany	Germany	\$1,983,988.04	2.47%	100.00%	18.25%
		Total	\$1,983,988.04	2.47%		18.25%
	United Kingdom	United Kingdom	\$4,279,008.83	5.32%	100.00%	39.36%
		Total	\$4,279,008.83	5.32%		39.36%
	Total		\$10,870,534.80	13.51%		100.00%
	North America	Canada	Canada	\$14,377,925.60	17.87%	100.00%
Total			\$14,377,925.60	17.87%		21.15%
United States		Central	\$7,906,008.18	9.83%	14.75%	11.63%
		Northeast	\$6,932,842.01	8.62%	12.93%	10.20%
		Northwest	\$12,435,076.00	15.46%	23.20%	18.29%
		Southeast	\$7,867,416.23	9.78%	14.68%	11.57%
		Southwest	\$18,466,458.79	22.95%	34.45%	27.16%
		Total	\$53,607,801.21	66.63%		78.85%
Total		\$67,985,726.81	84.51%		100.00%	
Pacific	Australia	Australia	\$1,594,335.38	1.98%	100.00%	100.00%
		Total	\$1,594,335.38	1.98%		100.00%
	Total		\$1,594,335.38	1.98%		100.00%
Total			\$80,450,596.98	100.00%		100.00%

En el objeto visual de matriz, observe que los valores de **Revenue % Total Country** ahora solo se muestran cuando una región está en el ámbito.


Realizar la transición de contexto

¿Qué ocurre cuando se evalúa una expresión de medida o una medida en el contexto de fila? Este escenario puede producirse en una fórmula de columna calculada o cuando se evalúa una expresión en una función de iterador.

En el ejemplo siguiente, agregará una columna calculada a la tabla **Customer** para clasificar los clientes en una clase de fidelidad. El escenario es sencillo: cuando los ingresos producidos por el cliente son inferiores a 2500 \$, el cliente se clasifica como **Bajo**; de lo contrario, se clasifica como **Alto**.

```
Customer Segment =
VAR CustomerRevenue = SUM(Sales[Sales Amount])
RETURN
    IF(CustomerRevenue < 2500, "Low", "High")
```

En la **Página 4** del informe, agregue la columna **Customer Segment** como la leyenda del gráfico circular.




Tenga en cuenta que solo existe un valor **Customer Segment**. La razón es que la fórmula de la columna calculada genera un resultado incorrecto: A cada cliente se le asigna el valor **Alto** porque la expresión `SUM(Sales[Sales Amount])` no se evalúa en un contexto de filtro. Por consiguiente, cada cliente se evalúa en la suma de *cada* columna **Importe de venta** en la tabla **Sales**.

Para forzar la evaluación de la expresión `SUM(Sales[Sales Amount])` *para cada cliente*, debe tener lugar una transacción de contexto que aplique los valores de la columna de contexto de fila al contexto de filtro. Puede realizar esta transacción mediante la función `CALCULATE` sin pasar expresiones de filtro.

Modifique la definición de la columna calculada para que genere el resultado correcto.

```
Customer Segment =  
VAR CustomerRevenue = CALCULATE(SUM(Sales[Sales Amount]))  
RETURN  
    IF(CustomerRevenue < 2500, "Low", "High")
```

En el objeto visual de gráfico circular, agregue la nueva columna calculada al cuadro de Leyenda y compruebe que se muestren dos segmentos circulares.



En este caso, la función CALCULATE aplica los valores de contexto de fila como filtros, conocidos como *transición de contexto*. Para ser precisos, el proceso no funciona de la misma manera cuando en la tabla hay una sola columna. Cuando hay una sola columna en la tabla, solo tiene que aplicar un filtro en esa columna para que se produzca la transición. En este caso, Power BI aplica un filtro en la columna **CustomerKey** para el valor en el contexto de fila.

Si hace referencia a medidas en una expresión que se evalúa en el contexto de fila, la transición de contexto es automática. Por lo tanto, no es necesario pasar referencias de medida a la función CALCULATE.

Modifique la definición de la columna calculada, que hace referencia a la medida **Revenue** y observe que continúa generando el resultado correcto.

```
Customer Segment =  
VAR CustomerRevenue = [Revenue]  
RETURN  
    IF(CustomerRevenue < 2500, "Low", "High")
```